

# NAG Fortran Library Routine Document

## F08FUF (CUNMTR/ZUNMTR)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08FUF (CUNMTR/ZUNMTR) multiplies an arbitrary complex matrix  $C$  by the complex unitary matrix  $Q$  which was determined by F08FSF (CHETRD/ZHETRD) when reducing a complex Hermitian matrix to tridiagonal form.

### 2 Specification

```

SUBROUTINE F08FUF(SIDE, UPLO, TRANS, M, N, A, LDA, TAU, C, LDC, WORK,
1                LWORK, INFO)
ENTRY          cunmtr (SIDE, UPLO, TRANS, M, N, A, LDA, TAU, C, LDC, WORK,
1                LWORK, INFO)
INTEGER        M, N, LDA, LDC, LWORK, INFO
complex      A(LDA,*), TAU(*), C(LDC,*), WORK(*)
CHARACTER*1    SIDE, UPLO, TRANS

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine is intended to be used after a call to F08FSF (CHETRD/ZHETRD), which reduces a complex Hermitian matrix  $A$  to real symmetric tridiagonal form  $T$  by a unitary similarity transformation:  $A = QTQ^H$ . F08FSF represents the unitary matrix  $Q$  as a product of elementary reflectors.

This routine may be used to form one of the matrix products

$$QC, Q^H C, CQ \text{ or } CQ^H,$$

overwriting the result on  $C$  (which may be any complex rectangular matrix).

A common application of this routine is to transform a matrix  $Z$  of eigenvectors of  $T$  to the matrix  $QZ$  of eigenvectors of  $A$ .

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: SIDE – CHARACTER\*1 *Input*

*On entry:* indicates how  $Q$  or  $Q^H$  is to be applied to  $C$  as follows:

if SIDE = 'L',  $Q$  or  $Q^H$  is applied to  $C$  from the left;

if SIDE = 'R',  $Q$  or  $Q^H$  is applied to  $C$  from the right.

*Constraint:* SIDE = 'L' or 'R'.

2: UPLO – CHARACTER\*1 *Input*

*On entry:* this **must** be the same parameter UPLO as supplied to F08FSF (CHETRD/ZHETRD).

*Constraint:* UPLO = 'U' or 'L'.

- 3: TRANS – CHARACTER\*1 *Input*  
*On entry:* indicates whether  $Q$  or  $Q^H$  is to be applied to  $C$  as follows:  
 if TRANS = 'N',  $Q$  is applied to  $C$ ;  
 if TRANS = 'C',  $Q^H$  is applied to  $C$ .  
*Constraint:* TRANS = 'N' or 'C'.
- 4: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $C$ ;  $m$  is also the order of  $Q$  if SIDE = 'L'.  
*Constraint:*  $M \geq 0$ .
- 5: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $C$ ;  $n$  is also the order of  $Q$  if SIDE = 'R'.  
*Constraint:*  $N \geq 0$ .
- 6: A(LDA,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, M)$  if SIDE = 'L' and at least  $\max(1, N)$  if SIDE = 'R'.  
*On entry:* details of the vectors which define the elementary reflectors, as returned by F08FSF (CHETRD/ZHETRD).  
*On exit:* used as internal workspace prior to being restored and hence is unchanged.
- 7: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08FUF (CUNMTR/ZUNMTR) is called.  
*Constraints:*  
 $LDA \geq \max(1, M)$  if SIDE = 'L',  
 $LDA \geq \max(1, N)$  if SIDE = 'R'.
- 8: TAU(\*) – **complex** array *Input*  
**Note:** the dimension of the array TAU must be at least  $\max(1, M - 1)$  if SIDE = 'L' and at least  $\max(1, N - 1)$  if SIDE = 'R'.  
*On entry:* further details of the elementary reflectors, as returned by F08FSF (CHETRD/ZHETRD).
- 9: C(LDC,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array C must be at least  $\max(1, N)$ .  
*On entry:* the  $m$  by  $n$  matrix  $C$ .  
*On exit:* C is overwritten by  $QC$  or  $Q^H C$  or  $CQ$  or  $CQ^H$  as specified by SIDE and TRANS.
- 10: LDC – INTEGER *Input*  
*On entry:* the first dimension of the array C as declared in the (sub)program from which F08FUF (CUNMTR/ZUNMTR) is called.  
*Constraint:*  $LDC \geq \max(1, M)$ .
- 11: WORK(\*) – **complex** array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, LWORK)$ .  
*On exit:* if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimum performance.

## 12: LWORK – INTEGER

*Input*

*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08FUF (CUNMTR/ZUNMTR) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).

*Suggested value:* for optimum performance LWORK should be at least  $N \times nb$  if SIDE = 'L' and at least  $M \times nb$  if SIDE = 'R', where *nb* is the **blocksize**.

*Constraints:*

$$\begin{aligned} \text{LWORK} &\geq \max(1, N) \text{ or } \text{LWORK} = -1 \text{ if } \text{SIDE} = \text{'L'}, \\ \text{LWORK} &\geq \max(1, M) \text{ or } \text{LWORK} = -1 \text{ if } \text{SIDE} = \text{'R'}. \end{aligned}$$

## 13: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = -*i*, the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed result differs from the exact result by a matrix *E* such that

$$\|E\|_2 = O(\epsilon)\|C\|_2,$$

where  $\epsilon$  is the *machine precision*.

## 8 Further Comments

The total number of real floating-point operations is approximately  $8m^2n$  if SIDE = 'L' and  $8mn^2$  if SIDE = 'R'.

The real analogue of this routine is F08FGF (SORMTR/DORMTR).

## 9 Example

To compute the two smallest eigenvalues, and the associated eigenvectors, of the matrix *A*, where

$$A = \begin{pmatrix} -2.28 + 0.00i & 1.78 - 2.03i & 2.26 + 0.10i & -0.12 + 2.53i \\ 1.78 + 2.03i & -1.12 + 0.00i & 0.01 + 0.43i & -1.07 + 0.86i \\ 2.26 - 0.10i & 0.01 - 0.43i & -0.37 + 0.00i & 2.31 - 0.92i \\ -0.12 - 2.53i & -1.07 - 0.86i & 2.31 + 0.92i & -0.73 + 0.00i \end{pmatrix}.$$

Here *A* is Hermitian and must first be reduced to tridiagonal form *T* by F08FSF (CHETRD/ZHETRD). The program then calls F08JJF (SSTEBZ/DSTEBZ) to compute the requested eigenvalues and F08JXF (CSTEIN/ZSTEIN) to compute the associated eigenvectors of *T*. Finally F08FUF (CUNMTR/ZUNMTR) is called to transform the eigenvectors to those of *A*.

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08FUF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LDZ, LWORK
PARAMETER        (NMAX=8,LDA=NMAX,LDZ=NMAX,LWORK=64*NMAX)
real           ZERO
PARAMETER        (ZERO=0.0e0)
*      .. Local Scalars ..
real          VL, VU
INTEGER          I, IFAIL, INFO, J, M, N, NSPLIT
CHARACTER        UPLO
*      .. Local Arrays ..
complex       A(LDA,NMAX), TAU(NMAX), WORK(LWORK), Z(LDZ,NMAX)
real          D(NMAX), E(NMAX), RWORK(5*NMAX), W(NMAX)
INTEGER          IBLOCK(NMAX), IFAILV(NMAX), ISPLIT(NMAX),
+              IWORK(NMAX)
CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL         sstebz, X04DBF, chetrd, cstein, cunmtr
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08FUF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) UPLO
IF (UPLO.EQ.'U') THEN
  READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.'L') THEN
  READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
END IF
*
*      Reduce A to tridiagonal form T = (Q**H)*A*Q
*
CALL chetrd(UPLO,N,A,LDA,D,E,TAU,WORK,LWORK,INFO)
*
*      Calculate the two smallest eigenvalues of T (same as A)
*
CALL sstebz('I','B',N,VL,VU,1,2,ZERO,D,E,M,NSPLIT,W,IBLOCK,
+          ISPLIT,RWORK,IWORK,INFO)
*
WRITE (NOUT,*)
IF (INFO.GT.0) THEN
  WRITE (NOUT,*) 'Failure to converge.'
ELSE
  WRITE (NOUT,*) 'Eigenvalues'
  WRITE (NOUT,99999) (W(I),I=1,M)
*
*      Calculate the eigenvectors of T, storing the result in Z
*
CALL cstein(N,D,E,M,W,IBLOCK,ISPLIT,Z,LDZ,RWORK,IWORK,
+          IFAILV,INFO)
*
IF (INFO.GT.0) THEN
  WRITE (NOUT,*) 'Failure to converge.'
ELSE
*
*      Calculate the eigenvectors of A = Q * (eigenvectors of T)
*
CALL cunmtr('Left',UPLO,'No transpose',N,M,A,LDA,TAU,Z,

```

```

+          LDZ,WORK,LWORK,INFO)
*
*          Print eigenvectors
*
*          WRITE (NOUT,*)
*          IFAIL = 0
*
*          CALL X04DBF('General',' ',N,M,Z,LDZ,'Bracketed','F7.4',
+                    'Eigenvectors','Integer',RLABS,'Integer',
+                    CLABS,80,0,IFAIL)
*
*          END IF
*          END IF
*          END IF
*          STOP
*
*          99999 FORMAT (8X,4(F7.4,11X,:))
*          END

```

## 9.2 Program Data

F08FUF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
(-2.28, 0.00)
( 1.78, 2.03) (-1.12, 0.00)
( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
(-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00) :End of matrix A

```

## 9.3 Program Results

F08FUF Example Program Results

```

Eigenvalues
-6.0002          -3.0030

```

```

Eigenvectors
          1          2
1 ( 0.7299, 0.0000) (-0.2595, 0.0000)
2 (-0.1663,-0.2061) ( 0.5969, 0.4214)
3 (-0.4165,-0.1417) (-0.2965,-0.1507)
4 ( 0.1743, 0.4162) ( 0.3482, 0.4085)

```

---